LAB GOALS

- To better understand the use of variables, data types and arithmetic and logical operators, conditionals and loops.
- To develop a simple Credit Card Payment Calculator.

In the process of building this lab, you will learn about: Console Applications, variable declaration (Dim), Data types (integer, double) arithmetic operators, conditionals [if-then-else], type conversion functions [CDbl( )], and Compiler directives such as Option Strict, Option Explicit.

Problem Statement:
Access to credit cards provides financial flexibility to many college students. However, many may not completely understand the real cost of obtaining credit cards. This lab will help you understand the relationship between “compound interest calculation” and the “minimum payment”. Unfortunately, many students choose to pay only the minimum payment on their credit cards.

After completing this lab you should have a better understanding of the disadvantages of paying only the minimum payment, and should be able to answer the following question: How many months does it take to payoff my credit card if I only pay the minimum payment on my bill?

Analysis:

<table>
<thead>
<tr>
<th>What is my credit card balance?</th>
<th>Ask the user. (Let’s say we charged $1000 to this card )</th>
</tr>
</thead>
<tbody>
<tr>
<td>What is my credit card annual interest rate?</td>
<td>Ask the user. (18% is a common interest rate, but some cards are as high as 29%)</td>
</tr>
<tr>
<td>How is the Minimum Payment calculated?</td>
<td>=&gt; 1% of the loan amount, plus the interest for the month. =&gt; $10 payment, if “minimum payment” calculation is &lt;= $10. =&gt; Full balance of the loan, if the balance is &lt;= $10.</td>
</tr>
<tr>
<td>What is my monthly interest rate?</td>
<td>Monthly Interest Rate is calculated by taking the Annual Interest Rate and dividing it by 12.</td>
</tr>
</tbody>
</table>

Design:

```
Print "What is your credit card balance?"
Input InitialBalance
Print "What is your credit card annual interest rate? (e.g. type 18 for 18%)"
Input AnnualIntRate
AnnualIntRate = AnnualIntRate / 100 ' Convert interest rate to Percent format
MonthlyIntRate = AnnualIntRate / 12
BeginOfMonthBalance = InitialBalance
While (BeginOfMonthBalance > 0)
    PrincipalThisMonth = BeginOfMonthBalance * (1 / 100) ' 1% of the loan
    InterestThisMonth = BeginOfMonthBalance * MonthlyIntRate
    MinimumPayment = PrincipalThisMonth + InterestThisMonth ' Regular minimum balance calculation
    If (MinimumPayment <= 10.0) Then
        MinimumPayment = 10.0 ' Minimum payment should be $10 or the full balance
        PrincipalThisMonth = MinimumPayment - InterestThisMonth
    End If
    If (BeginOfMonthBalance <= 10.0) Then
        MinimumPayment = BeginOfMonthBalance ' Pay the full balance
        InterestThisMonth = 0
    End If
    EndOfMonthBalance = BeginOfMonthBalance - PrincipalThisMonth
    Print BeginOfMonthBalance, InterestThisMonth, PrincipalThisMonth, MinimumPayment, EndOfMonthBalance
    BeginOfMonthBalance = EndOfMonthBalance ' Calculate the beginning balance for the next month
End While
```
Implementation:

Step 1: To get started, proceed to open up Microsoft Visual Basic .NET. Create a new Console Mode Project named “CreditCardPayments.” You should see the figure below.

Step 2: Add the following two lines at the beginning of the program. (Above the “Module Module1” line:

```
Option Explicit On
Option Strict On
```

Question?
Do you remember what these two line do? If not, review your notes to find the answer before continuing with the lab.

Right after the “Sub Main()” statement, add the following variable declarations to your program:

```
Dim InitialBalance, AnnualIntRate, MonthlyIntRate As Double
Dim InterestThisMonth, PrincipalThisMonth As Double
Dim BeginOfMonthBalance, EndOfMonthBalance As Double
Dim MinimumPayment As Double
Dim payment_no As Integer
```

Run: Compile and Run your program. If your program does not compile, fix the syntax errors and compile the program again. Once you are able to successfully run the program, you should see that the program quickly executes and closes the Console window. In order to allow you better observe the output of the program, Right before the “End Sub” statement, add the following line in your program:

```
Console.ReadLine()                                'Pause the program
```

Run the program again. This time, the program should run and wait for your to hit the “ENTER” key before closing the console window. Note, at this point the program is not doing much, we are compiling and running it, just to make sure there are not syntactic errors in the program.
Step 3: Add the following lines of code between the variable declarations and the Console.ReadLine() statement:

```csharp
Console.WriteLine("What is your credit card balance?")
InitialBalance = CDbl(Console.ReadLine())

Console.WriteLine("What is your credit card annual interest rate? (enter whole numbers)"
AnnualIntRate = CDbl(Console.ReadLine())
AnnualIntRate = AnnualIntRate / 100         ' Convert interest rate to Percent format
MonthlyIntRate = AnnualIntRate / 12

BeginOfMonthBalance = InitialBalance

Console.WriteLine("PmtNo    BegBal    IntThisMonth  PrincThisMonth  Payment     EndBal")
While (BeginOfMonthBalance > 0)
    PrincipalThisMonth = BeginOfMonthBalance * (1 / 100)
    InterestThisMonth = BeginOfMonthBalance * MonthlyIntRate
    MinimumPayment = PrincipalThisMonth + InterestThisMonth  ' Regular minimum balance calculation
    If (MinimumPayment <= 10.0) Then
        MinimumPayment = 10.0                    ' Minimum payment should be $10 or the full balance
        PrincipalThisMonth = MinimumPayment - InterestThisMonth
    End If
    If (BeginOfMonthBalance <= 10.0) Then
        MinimumPayment = BeginOfMonthBalance     ' Pay the full balance
        PrincipalThisMonth = BeginOfMonthBalance
        InterestThisMonth = 0
    End If
    EndOfMonthBalance = BeginOfMonthBalance - PrincipalThisMonth
    Console.WriteLine("{0} {1} {2} {3}".BeginOfMonthBalance, InterestThisMonth, PrincipalThisMonth, EndOfMonthBalance)
    BeginOfMonthBalance = EndOfMonthBalance       ' Calculate the beginning balance for the next month
End While
```

Run: Compile and Run your code again to see if it runs properly. You should see four columns of numbers. The first column represents your balance at the beginning of a given month. The second column shows how much interest you are paying that month, and the third column shows how much of your payment is going toward the principal of the loan. And finally, the last column shows how much you owe at the end of the month.
Step 4: Well, the columns do not look very pretty do they? Well, let's fix that. Replace the Console.WriteLine() statement which displays the columns with the following line:

```
Console.WriteLine("{0,12:c} {1,12:c} {2,12:c} {3,12:c}", BeginOfYearBalance, InterestThisMonth, PrincipalThisMonth, EndOfYearBalance)
```

The above WriteLine() statement includes some formatting in it. Here is what the `{0,12:c}` formatting means:

- The “0” means the first place holder which refers to the “BeginOfMonthBalance” variable.
- The “,” is a separator
- The “12” indicates that the resulting value should be placed in 12 spaces.
- The “:c” indicates currency format. In other words the numbers will appear with a preceding $ sign and 2 digits after the decimal point.
Run: Run your program again to see if it runs properly. This time the columns should line up nicely.

Question? Lets ask the question again:

**How many months does it take to payoff my credit card if I only pay the minimum payment on my credit card bill?**

Step5: Complete the program (see below) and run it. You should see how many payments it takes to pay off $1000, $5000, etc. if you only pay the minimum payment.
Calculating Credit Card Payments

Programmer: Hossein Hakimzadeh
Date: 1/20/2008

Description: This project will calculate and display credit card payments. The program assumes that the borrower only pays the minimum payment on each bill. The minimum payment is calculated by adding 1% of the loan amount, and the monthly interest owed to the credit card company.

Option Strict On
Option Explicit On

Module Module1

Sub Main()
    Dim InitialBalance, AnnualIntRate, MonthlyIntRate As Double
    Dim InterestThisMonth, PrincipalThisMonth As Double
    Dim BeginOfMonthBalance, EndOfMonthBalance As Double
    Dim MinimumPayment As Double
    Dim payment_no As Integer

    Console.WriteLine("What is your credit card balance? ")
    InitialBalance = CDbl(Console.ReadLine())

    Console.WriteLine("What is your credit card annual interest rate (e.g. enter 18 for 18%)? ")
    AnnualIntRate = CDbl(Console.ReadLine())

    AnnualIntRate = AnnualIntRate / 100  ' Convert interest rate to Percent format
    MonthlyIntRate = AnnualIntRate / 12

    BeginOfMonthBalance = InitialBalance
    payment_no = 1

    Console.WriteLine("PmtNo    BegBal    IntThisMonth  PrincThisMonth  Payment     EndBal")
    While (BeginOfMonthBalance > 0)
        PrincipalThisMonth = BeginOfMonthBalance * (1 / 100)
        InterestThisMonth = BeginOfMonthBalance * MonthlyIntRate

        MinimumPayment = PrincipalThisMonth + InterestThisMonth  ' Regular minimum balance calculation
        If (MinimumPayment <= 10.0) Then
            MinimumPayment = 10.0
            PrincipalThisMonth = MinimumPayment - InterestThisMonth
        End If
        If (BeginOfMonthBalance <= 10.0) Then
            MinimumPayment = BeginOfMonthBalance
            PrincipalThisMonth = BeginOfMonthBalance
            InterestThisMonth = 0
        End If

        EndOfMonthBalance = BeginOfMonthBalance - PrincipalThisMonth

        Console.WriteLine("{0,3:d} {1,12:c} {2,12:c} {3,12:c} {4,12:c} {5,12:c}", payment_no, BeginOfMonthBalance, InterestThisMonth, PrincipalThisMonth, MinimumPayment, EndOfMonthBalance)
        BeginOfMonthBalance = EndOfMonthBalance
        payment_no = payment_no + 1
    End While

    Console.ReadLine()                                'Pause the program
End Sub

End Module